This is a part of SFQED-Loops script collection
developed for calculating loop and tree-level processes in
Strong-Field Quantum Electrodynamics.
The scripts are available on https://github.com/ArsenyMironov/SFQED-Loops

If you use this script in your research, please, consider
citing our papers:
- A. A. Mironov, S. Meuren, and A. M. Fedotov, PRD 102, 053005 (2020); arXiv:2003.06909
  https://doi.org/10.1103/PhysRevD.102.053005
- A. A. Mironov, A. M. Fedotov, 105 (3), 033005 (2022); arXiv:2109.00634
https://doi.org/10.1103/PhysRevD.105.033005
- Y. V. Selivanov, A. A. Mironov, A. I. Alexeenko, A. M. Fedotov arXiv:2605.12479 (2026)

If you have any questions, please, don’t hesitate to contact:
mironov.hep@gmail.com

Copyright (C) 2026 Arseny Mironov, https://scholar.google.ru/citations?user=GQwNEuwAAAAJ

This program is free software: you can redistribute it and/or modify
it under the terms of the GNU General Public License as published by
the Free Software Foundation, either version 3 of the License, or
(at your option) any later version.

This program is distributed in the hope that it will be useful,
but WITHOUT ANY WARRANTY; without even the implied warranty of
MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the
GNU General Public License for more details.

You should have received a copy of the GNU General Public License
along with this program. If not, see <https://www.gnu.org/licenses/>.



In[1]:=
```
NotebookEvaluate[NotebookOpen[NotebookDirectory[] <> "definitions.nb"]]
```

FeynCalc 10.1.0 (stable version). For help, use the online documentation, visit the forum and have a look at the supplied examples. The PDF–version of the manual can be downloaded here.

If you use FeynCalc in your research, please evaluate FeynCalcHowToCite[] to learn how to cite this software.

Please keep in mind that the proper academic attribution of our work is crucial to ensure the future development of this package!

Electron propagator (proper time is dimensionless)

$$S^c(x_2, x_1) = \Lambda^{4-D} \int \frac{d^D p}{(2\pi)^D} E_p(x_2) \frac{i((\gamma p) + m)}{p^2 - m^2 + i0} \bar{E}_p(x_1)$$

$x = x_2 - x_1$,

$X = \frac{1}{2}(x_1 + x_2)$,

$\xi^2 = -\frac{e^2 a^2}{m^2}$,

$[\Lambda] = m$ – mass scale,

$$E_p(x_2) = \left[1 - \frac{e(\gamma k)(\gamma a)}{2(kp)}(kx_2)\right]$$

$$\text{Exp}\left[-i(p\,x_2) + i\frac{e(a\,p)}{2(k\,p)}(k\,x_2)^2 + i\frac{a^2 e^2}{6(k\,p)}(k\,x_2)^3\right];$$

$(\gamma p - m) S^c(p) = i$ – in E – p representation

$S^c(p, q) = (2\pi)^4 \delta(p - q) S^c(p)$

In[2]:=
```
NewMomentum["p"]
NewCoordinate["x1"]
NewCoordinate["x2"]
NewCoordinate["x"]
NewCoordinate["X"]
```

$$\left\{p^\alpha, p^2, k\cdot p, \text{Fp}^\alpha, \text{FFp}^\alpha, \text{FDp}^\alpha, a\cdot p, 0, 0, 0, -a^2(k\cdot p), 0, 0, -\frac{m^6 \chi \text{p}^2}{e^2}, -\frac{m^6 \chi \text{p}^2}{e^2}, \frac{m^6 \chi \text{p}^2}{e^2}, 0, 0, 0, 0, 0, 0\right\}$$

$$\left\{\text{x1}^\alpha, \text{x1}^2, k\cdot \text{x1}, a\cdot \text{x1}, \text{Fx1}^\alpha, \text{FFx1}^\alpha, \text{FDx1}^\alpha, k\cdot \text{x1}, 0, 0, 0, -a^2(k\cdot \text{x1}), 0, 0, -\frac{m^2 \xi^2 (k\cdot \text{x1})^2}{e^2}, -\frac{m^2 \xi^2 (k\cdot \text{x1})^2}{e^2}, \frac{m^2 \xi^2 (k\cdot \text{x1})^2}{e^2}, 0, 0, 0, 0, 0, 0\right\}$$

$$\left\{\text{x2}^\alpha, \text{x2}^2, k\cdot \text{x2}, a\cdot \text{x2}, \text{Fx2}^\alpha, \text{FFx2}^\alpha, \text{FDx2}^\alpha, k\cdot \text{x2}, 0, 0, 0, -a^2(k\cdot \text{x2}), 0, 0, -\frac{m^2 \xi^2 (k\cdot \text{x2})^2}{e^2}, -\frac{m^2 \xi^2 (k\cdot \text{x2})^2}{e^2}, \frac{m^2 \xi^2 (k\cdot \text{x2})^2}{e^2}, 0, 0, 0, 0, 0, 0\right\}$$

$$\left\{x^\alpha, x^2, k\cdot x, a\cdot x, \text{Fx}^\alpha, \text{FFx}^\alpha, \text{FDx}^\alpha, k\cdot x, 0, 0, 0, -a^2(k\cdot x), 0, 0, -\frac{m^2 \xi^2 (k\cdot x)^2}{e^2}, -\frac{m^2 \xi^2 (k\cdot x)^2}{e^2}, \frac{m^2 \xi^2 (k\cdot x)^2}{e^2}, 0, 0, 0, 0, 0, 0\right\}$$

$$\left\{X^\alpha, X^2, k\cdot X, a\cdot X, \text{FX}^\alpha, \text{FFX}^\alpha, \text{FDX}^\alpha, k\cdot X, 0, 0, 0, -a^2(k\cdot X), 0, 0, -\frac{m^2 \xi^2 (k\cdot X)^2}{e^2}, -\frac{m^2 \xi^2 (k\cdot X)^2}{e^2}, \frac{m^2 \xi^2 (k\cdot X)^2}{e^2}, 0, 0, 0, 0, 0, 0\right\}$$

In[7]:=
```
Epx2 = Ep[x2, p]
EpBarx1 = EpC[x1, p]
```

Out[7]= $$\left\{1-\frac{e\,(k\cdot \mathrm{x2})\,(\gamma\cdot k).(\gamma\cdot a)}{2\,(k\cdot p)},\ \frac{a^2\,e^2\,(\,k\cdot \mathrm{x2}\,)^3}{6\,(k\cdot p)}+\frac{e\,(a\cdot p)\,(\,k\cdot \mathrm{x2}\,)^2}{2\,(k\cdot p)}-p\cdot \mathrm{x2}\right\}$$

Out[8]= $$\left\{1-\frac{e\,(k\cdot \mathrm{x1})\,(\gamma\cdot a).(\gamma\cdot k)}{2\,(k\cdot p)},\ -\frac{a^2\,e^2\,(\,k\cdot \mathrm{x1}\,)^3}{6\,(k\cdot p)}-\frac{e\,(a\cdot p)\,(\,k\cdot \mathrm{x1}\,)^2}{2\,(k\cdot p)}+p\cdot \mathrm{x1}\right\}$$

In[9]:=
```
Matrix = Epx2⟦1⟧.(GAD[α] × Pair[Momentum[p, D], LorentzIndex[α, D]] + m).EpBarx1⟦1⟧
Coeff = i Λ^(4 - D) / (2 π)^D / (pv2 - m^2)
Phase = Epx2⟦2⟧ + EpBarx1⟦2⟧
```

Out[9]= $$\left(1-\frac{e\,(k\cdot \mathrm{x2})\,(\gamma\cdot k).(\gamma\cdot a)}{2\,(k\cdot p)}\right).(m+\gamma^{\alpha}\,p^{\alpha}).\left(1-\frac{e\,(k\cdot \mathrm{x1})\,(\gamma\cdot a).(\gamma\cdot k)}{2\,(k\cdot p)}\right)$$

Out[10]= $$\frac{i\,(2\,\pi)^{-D}\,\Lambda^{4-D}}{p^2-m^2}$$

Out[11]= $$-\frac{a^2\,e^2\,(\,k\cdot \mathrm{x1}\,)^3}{6\,(k\cdot p)}+\frac{a^2\,e^2\,(\,k\cdot \mathrm{x2}\,)^3}{6\,(k\cdot p)}-\frac{e\,(a\cdot p)\,(\,k\cdot \mathrm{x1}\,)^2}{2\,(k\cdot p)}+\frac{e\,(a\cdot p)\,(\,k\cdot \mathrm{x2}\,)^2}{2\,(k\cdot p)}+p\cdot \mathrm{x1}-p\cdot \mathrm{x2}$$

In[12]:=
```
Matrix1 = Contract[DiracSimplify[Matrix]]
Coeff1 = Coeff;
Phase1 = Phase;
```

Out[12]= $$-\frac{a^2\,e^2\,\gamma\cdot k\,(k\cdot \mathrm{x1})\,(k\cdot \mathrm{x2})}{2\,(k\cdot p)}-\frac{e\,m\,(k\cdot \mathrm{x1})\,(\gamma\cdot a).(\gamma\cdot k)}{2\,(k\cdot p)}-\frac{e\,m\,(k\cdot \mathrm{x2})\,(\gamma\cdot k).(\gamma\cdot a)}{2\,(k\cdot p)}-\frac{e\,(k\cdot \mathrm{x1})\,(\gamma\cdot p).(\gamma\cdot a).(\gamma\cdot k)}{2\,(k\cdot p)}-\frac{e\,(k\cdot \mathrm{x2})\,(\gamma\cdot k).(\gamma\cdot a).(\gamma\cdot p)}{2\,(k\cdot p)}+m+\gamma\cdot p$$

In[15]:=
```
Matrix2 = Expand[ExpandScalarProduct[
    Matrix1 /. {Momentum[x1, D] → Momentum[X, D] - Momentum[x, D] / 2,
      Momentum[x2, D] → Momentum[X, D] + Momentum[x, D] / 2}]]
Coeff2 = Coeff1;
Phase2 = Expand[ExpandScalarProduct[
    Phase1 /. {Momentum[x1, D] → Momentum[X, D] - Momentum[x, D] / 2,
      Momentum[x2, D] → Momentum[X, D] + Momentum[x, D] / 2}]]
```

Out[15]= $$\frac{a^2\,e^2\,\gamma\cdot k\,(\,k\cdot x\,)^2}{8\,(k\cdot p)}-\frac{a^2\,e^2\,\gamma\cdot k\,(\,k\cdot X\,)^2}{2\,(k\cdot p)}+\frac{e\,m\,(k\cdot x)\,(\gamma\cdot a).(\gamma\cdot k)}{4\,(k\cdot p)}-\frac{e\,m\,(k\cdot x)\,(\gamma\cdot k).(\gamma\cdot a)}{4\,(k\cdot p)}-\frac{e\,m\,(k\cdot X)\,(\gamma\cdot a).(\gamma\cdot k)}{2\,(k\cdot p)}-\frac{e\,m\,(k\cdot X)\,(\gamma\cdot k).(\gamma\cdot a)}{2\,(k\cdot p)}-\frac{e\,(k\cdot x)\,(\gamma\cdot k).(\gamma\cdot a).(\gamma\cdot p)}{4\,(k\cdot p)}+\frac{e\,(k\cdot x)\,(\gamma\cdot p).(\gamma\cdot a).(\gamma\cdot k)}{4\,(k\cdot p)}-\frac{e\,(k\cdot X)\,(\gamma\cdot k).(\gamma\cdot a).(\gamma\cdot p)}{2\,(k\cdot p)}-\frac{e\,(k\cdot X)\,(\gamma\cdot p).(\gamma\cdot a).(\gamma\cdot k)}{2\,(k\cdot p)}+m+\gamma\cdot p$$

Out[17]= $$\frac{a^2\,e^2\,(k\cdot x)\,(\,k\cdot X\,)^2}{2\,(k\cdot p)}+\frac{a^2\,e^2\,(\,k\cdot x\,)^3}{24\,(k\cdot p)}+\frac{e\,(a\cdot p)\,(k\cdot x)\,(k\cdot X)}{k\cdot p}-p\cdot x$$

In[18]:=
```
Matrix3 =
  (((Expand[Matrix2 /. TripleGamma]) /. EpsToF) /. FieldSubstitutions) /. akToF /.
   {DiracGamma[LorentzIndex[β, D], D].DiracGamma[5] ×
      Pair[LorentzIndex[β, D], Momentum[FDp, D]] → FVD[γγ5FD, α] × pv[α]}
Coeff3 = Coeff2;
Phase3 = Phase2;
```

Out[18]=
$$\frac{a^2 e^2 \gamma\cdot k\,(k\cdot x)^2}{8\,(k\cdot p)} - \frac{a^2 e^2 \gamma\cdot k\,(k\cdot X)^2}{2\,(k\cdot p)} - \frac{e\,(a\cdot p)\,\gamma\cdot k\,(k\cdot X)}{k\cdot p} + e\,\gamma\cdot a\,(k\cdot X) + \frac{i\,e\,m\,\sigma \mathrm{F}\,(k\cdot x)}{4\,(k\cdot p)} + \frac{i\,e\,\gamma\gamma5\mathrm{FD}^{\alpha}\,p^{\alpha}\,(k\cdot x)}{2\,(k\cdot p)} + m + \gamma\cdot p$$

Expanding scalar products into components and changing variables

$p \to \{p_- = 1/2\,(p^0 - p^3),\ p_+ = p^0 + p^3,\ p_\perp\}$

$p_- = x_- / 2\,s$

$p_+ = (p^2 - p_\perp{}^2) / 2\,p_- = s\,(p^2 + p_\perp{}^2) / x_-$

Integration measure

$\int d^D \mathbf{p} \ldots = \int \frac{ds}{2\,s}\, dp^2\, d^{D-2} p_\perp$

$x_- = kx / m$

$p_- = kx / 2\,m\,s$

$kp = m\,p_- = m\,xm / 2\,s = kx / 2\,s$

$ap = -at\,pt$

$\gamma p = \gamma_-\, p_+ + \gamma_+\, p_- - \gamma_\perp\, p_\perp = Gm\,\frac{s}{x_-}\,(p^2 + p_\perp{}^2) + Gp\,\frac{x_-}{2\,s} - Gt\,pt$

$\gamma k = \gamma_-\, k_+ = m\,Gm$

$kx = k_+\, x_- = m\,xm$

$(\gamma F^*)^\mu.\gamma^5 \to \{(\gamma F^*)_-.\gamma^5 = 0,\ (\gamma F^*)_+.\gamma^5,\ (\gamma F^*)_\perp.\gamma^5\} = \{0,\ \gamma\gamma5FDp,\ \gamma\gamma5FDt\}$

$(\gamma F^*)_\mu\, k^\mu = 0 \to (\gamma F^*)_- = 0$

$(\gamma F^*)_\mu\, a^\mu = 0 \to \gamma\gamma5FDt\ at = 0$

$(\gamma F^*)^\mu.\gamma^5\, p_\mu = \gamma\gamma5FDm * pp - \gamma\gamma5FDt * pt$

In[21]:=
```
D[{xm / 2 / s, s (p2 + pt2) / xm}, {{s, p2}}]
Abs[Det[%]]
```

Out[21]=
$$\begin{pmatrix} -\frac{xm}{2\,s^2} & 0 \\ \frac{p2+pt2}{xm} & \frac{s}{xm} \end{pmatrix}$$

Out[22]=
$$\frac{1}{2\,|s|}$$

In[23]:=
```
Matrix4 = Collect[
   Expand[Matrix3 /. {DiracGamma[Momentum[p, D], D] → Gp * pm + Gm * pp - Gt * pt,
        Pair[Momentum[a, D], Momentum[p, D]] → -at pt, FVD[γγ5FD, α_] × pv[α_] →
         γγ5FDp * pm - γγ5FDt * pt, DiracGamma[Momentum[k, D], D] → m Gm} /.
       {kp → kx / 2 / s, pm → xm / 2 / s} /. {kx → m xm} /.
     {pp → s (p2 + pt^2) / xm}], {pt, p2}]
Coeff4 = Coeff3 / 2 / s
Phase4 = Collect[
   Expand[Phase3 /. {Pair[Momentum[p, D], Momentum[x, D]] → pp xm + pm xp - pt * xt,
        kp → m pm, Pair[Momentum[a, D], Momentum[p, D]] → -at pt } /.
       {kp → kx / 2 / s, pm → xm / 2 / s} /. {pp → s (p2 + pt^2) / xm} /. {xm →
       kx / m}], {pt, pp}]
```

Out[23]= $-\frac{a^2 e^2 \mathrm{Gm}\, s\, (k\cdot X)^2}{\mathrm{xm}} + \frac{1}{4} a^2 e^2 \mathrm{Gm}\, m^2 s\, \mathrm{xm} + e\, \gamma\cdot a\, (k\cdot X) + \mathrm{pt}\left(\frac{2\, \mathrm{at}\, e\, \mathrm{Gm}\, s\, (k\cdot X)}{\mathrm{xm}} - i\, \gamma\gamma 5\mathrm{FDt}\, e\, s - \mathrm{Gt}\right) + \frac{1}{2} i\, e\, m\, s\, \sigma\mathrm{F} + \frac{1}{2} i\, \gamma\gamma 5\mathrm{FDp}\, e\, \mathrm{xm} + \frac{\mathrm{Gm}\, \mathrm{p2}\, s}{\mathrm{xm}} + \frac{\mathrm{Gm}\, \mathrm{pt}^2 s}{\mathrm{xm}} + \frac{\mathrm{Gp}\, \mathrm{xm}}{2 s} + m$

Out[24]= $\frac{i\, 2^{-D-1} \pi^{-D} \Lambda^{4-D}}{s\left(p^2 - m^2\right)}$

Out[25]= $\frac{1}{12} a^2 e^2 s\, (k\cdot x)^2 + a^2 e^2 s\, (k\cdot X)^2 + \mathrm{pt}\, (\mathrm{xt} - 2\, \mathrm{at}\, e\, s\, (k\cdot X)) - \frac{\mathrm{xp}\, (k\cdot x)}{2\, m\, s} - \mathrm{p2}\, s + \mathrm{pt}^2 (-s)$

Integration over

$\int d^{D-2} p_\perp \ldots$

$$I_0 = \int d^{D-2} p_\perp \, \text{Exp}\left[-I\, A\, p_\perp^2 + I\, (J.p_\perp)\right] =$$

$$= \text{Exp}\left[-I \frac{\pi}{2} \frac{D-2}{2}\right] \pi^{\frac{D-2}{2}} (\det A)^{-\frac{1}{2}} \, \text{Exp}\left[I \frac{1}{4} J.A^{-1}.J\right]$$

$$I_1 = \int d^{D-2} p_\perp \, p_\perp \, \text{Exp}\left[-I\, A\, p_\perp^2 + I\, (J.p_\perp)\right] =$$

$$= \frac{1}{2} A^{-1}.J\, I_0$$

$$I_2 = \int d^{D-2} p_\perp \, p_\perp{}^2 \, \text{Exp}\left[-I\, A\, p_\perp^2 + I\, (J.p_\perp)\right] =$$

$$= \left[-i \frac{1}{2} \text{Tr}\, A^{-1} + \left(\frac{1}{2} A^{-1}.J\right)^2\right] I_0$$

where

$A = s,$

$J = x_\perp - 2\, ea_\perp\, s\, kX,$

$\det A = s^{D-2},$

$A^{-1} = 1/s$

We perform integrations

Integrations changes the coefficient (Coeff) and phase

Then recollect some scalar products

$a_\perp{}^\wedge 2 = -a^2$

$a_\perp\, x_\perp = -(ax)$

$x_\perp^2 = 2\, x_-\, x_+ - x^2$

In[26]:=
```
Amatr = -Coefficient[Phase4, pt^2]
J = Coefficient[Phase4, pt]
CI0 = Exp[-I Pi / 2 (D / 2 - 1)] Pi^(D / 2 - 1) / Amatr^(D / 2 - 1)
```

Out[26]= s

Out[27]= $\text{xt} - 2\, \text{at}\, e\, s\, (k \cdot X)$

Out[28]= $e^{-\frac{1}{2} i \pi \left(\frac{D}{2}-1\right)} \pi^{\frac{D}{2}-1} s^{1-\frac{D}{2}}$

In[29]:=
```
Phase5 = Expand[Expand[((Phase4 /. {pt → 0}) + 1 / 4 J^2 / Amatr)] /.
    { at^2 → -av2, at xt → -ax, xt^2 → 2 xm xp - xv2} /. {xm → kx / m}]
Coeff5 = Coeff4 * CI0
Matrix5 = Collect[
  Expand[Expand[(Matrix4 /. {pt → 0}) + Coefficient[Matrix4, pt] * 1 / 2 / Amatr * J +
      Coefficient[Matrix4, pt^2] * (-I / 2 * (D - 2) / Amatr + (1 / 2 / Amatr * J) ^2)] /.
    {γγ5FDt at → 0, at^2 → -av2, at xt → -ax, xt^2 → 2 xm xp - xv2}],
  {p2, Gm, Gp, Gt, γγ5FDm, γγ5FDp, γγ5FDt}]
```

Out[29]= $\frac{1}{12} a^2 e^2 s (k\cdot x)^2 + e (a\cdot x)(k\cdot X) - \text{p2}\, s - \frac{x^2}{4 s}$

Out[30]= $\frac{i\, 2^{-D-1} e^{-\frac{1}{2} i \pi \left(\frac{D}{2}-1\right)} \pi^{-\frac{D}{2}-1} \Lambda^{4-D} s^{-D/2}}{p^2 - m^2}$

Out[31]= $\text{Gm}\left(\frac{1}{4} a^2 e^2 m^2 s\, \text{xm} - \frac{i D}{2\, \text{xm}} - \frac{x^2}{4 s\, \text{xm}} + \frac{\text{xp}}{2 s} + \frac{i}{\text{xm}}\right) + e\, \gamma\cdot a (k\cdot X) + \text{Gt}\left(\text{at}\, e (k\cdot X) - \frac{\text{xt}}{2 s}\right) + \frac{1}{2} i e m s \sigma\text{F} + \frac{1}{2} i \gamma\gamma5\text{FDp}\, e\, \text{xm} - \frac{1}{2} i \gamma\gamma5\text{FDt}\, e\, \text{xt} + \frac{\text{Gm}\, \text{p2}\, s}{\text{xm}} + \frac{\text{Gp}\, \text{xm}}{2 s} + m$

Next we substitute

$\gamma_\perp a_\perp = -\gamma a$

$\gamma_\perp x_\perp = \gamma_- x_+ - \gamma_+ x_- - \gamma x$

$\gamma_- x_- = \frac{\gamma k}{m} x_-$

$x_- = kx / m$

In[32]:=
```
Matrix6 = Collect[Expand[Matrix5] /. {Gt at → -Contract[GAD[α] × av[α]],
      Gt xt → Gm xp + Gp xm - Contract[GAD[α] × xv[α]] ,
      γγ5FDt xt → γγ5FDp xm - DiracSlash[FDxv, Dimension → D].GA[5]} /.
    {Gm xm → DiracGamma[Momentum[k, D], D] / m * xm} /. {xm → kx / m},
  {p2, Gm, Gp, Gt, γγ5FDm, γγ5FDp, γγ5FDt}]
Coeff6 = Coeff5
Phase6 = Phase5
```

Out[32]= $\frac{1}{2} i e (\gamma\cdot\text{FDxv}).\bar{\gamma}^5 + \frac{1}{4} a^2 e^2 s\, \gamma\cdot k (k\cdot x) + \text{Gm}\left(-\frac{i D m}{2 (k\cdot x)} - \frac{m x^2}{4 s (k\cdot x)} + \frac{i m}{k\cdot x}\right) + \frac{1}{2} i e m s \sigma\text{F} + \frac{\text{Gm}\, m\, \text{p2}\, s}{k\cdot x} + m + \frac{\gamma\cdot x}{2 s}$

Out[33]= $\frac{i\, 2^{-D-1} e^{-\frac{1}{2} i \pi \left(\frac{D}{2}-1\right)} \pi^{-\frac{D}{2}-1} \Lambda^{4-D} s^{-D/2}}{p^2 - m^2}$

Out[34]= $\frac{1}{12} a^2 e^2 s (k\cdot x)^2 + e (a\cdot x)(k\cdot X) - \text{p2}\, s - \frac{x^2}{4 s}$

$\nu = p^2 - m^2$

$\int \frac{d\nu}{\nu + i0} \exp[-i s \nu] = -2 \pi i\, \theta(s)$

$\int d\nu \exp[-i s \nu] = 2 \pi \delta(s)$

$$\int_{-\infty}^{\infty} \frac{ds}{s^{D/2}}\, s\, e^{i g(s)}\, 2\pi\,\delta(s) =$$

$$-2\pi i \int_{-\infty}^{\infty} \frac{ds}{s^{D/2}}\, e^{i g(s)}\, \theta(s)\,[i\,(D/2-1) + s\, g'(s)]$$

where

$g(s) = \text{Phase6 with } p^2 = m^2$

Finally,

$Gm = \gamma_- = \gamma k / m$

In[35]:=
```
Phase7 = Phase6 /. {p2 → m^2}
Coeff7 = Coeff6 * (pv2 - m^2) * (-2 π i)
g = Phase7;
Matrix7 = Expand[
   (Matrix6 /. {p2 → m^2}) + Coefficient[Matrix6, p2 s] * (i (D / 2 - 1) + s D[g, s])] /.
  {Gm → DiracGamma[Momentum[k, D], D] / m}
```

Out[35]= $\frac{1}{12} a^2 e^2 s\,(k\cdot x)^2 + e\,(a\cdot x)\,(k\cdot X) + m^2\,(-s) - \frac{x^2}{4 s}$

Out[36]= $2^{-D}\, e^{-\frac{1}{2} i \pi \left(\frac{D}{2}-1\right)}\, \pi^{-D/2}\, \Lambda^{4-D}\, s^{-D/2}$

Out[38]= $\frac{1}{2} i\, e\,(\gamma\cdot \text{FDxv}).\bar{\gamma}^5 + \frac{1}{3} a^2 e^2 s\, \gamma\cdot k\,(k\cdot x) + \frac{1}{2} i\, e\, m\, s\, \sigma\text{F} + m + \frac{\gamma\cdot x}{2 s}$

In[39]:=
```
Matrix8 = Expand[Matrix7 / m /.
   {DiracGamma[Momentum[k, D], D] → Contract[GAD[α] × FFxv[α]] / (-av2 kx)}]
Coeff8 = Coeff7 * m
Phase8 = Phase7 /. {av2 kx^2 → Fx^2}
```

Out[39]= $\frac{i\, e\,(\gamma\cdot \text{FDxv}).\bar{\gamma}^5}{2 m} - \frac{e^2 s\, \gamma\cdot \text{FFx}}{3 m} + \frac{1}{2} i\, e\, s\, \sigma\text{F} + \frac{\gamma\cdot x}{2 m s} + 1$

Out[40]= $2^{-D}\, e^{-\frac{1}{2} i \pi \left(\frac{D}{2}-1\right)}\, \pi^{-D/2}\, m\, \Lambda^{4-D}\, s^{-D/2}$

Out[41]= $e\,(a\cdot x)\,(k\cdot X) + \frac{1}{12} e^2\, \text{Fx}^2\, s - m^2 s - \frac{x^2}{4 s}$

In[42]:=
```
Matrix9 = Matrix8 /. {s → s1 / m^2}
Coeff9 = Simplify[Coeff8 / m^2 /. {s → s1 / m^2}, Assumptions → m > 0]
Coeff9 /. {D → 4}
Phase9 = Phase8 /. {s → s1 / m^2}
```

Out[42]= $$\frac{i\, e\, (\gamma\cdot \mathrm{FDxv}).\bar{\gamma}^5}{2\, m} - \frac{e^2\, \mathrm{s1}\, \gamma\cdot \mathrm{FFx}}{3\, m^3} + \frac{i\, e\, \mathrm{s1}\, \sigma\mathrm{F}}{2\, m^2} + \frac{m\, \gamma\cdot x}{2\, \mathrm{s1}} + 1$$

Out[43]= $$i\, 2^{-D}\, e^{-\frac{1}{4} i \pi D}\, \pi^{-D/2}\, \Lambda^{4-D}\, m^{D-1}\, \mathrm{s1}^{-D/2}$$

Out[44]= $$-\frac{i\, m^3}{16\, \pi^2\, \mathrm{s1}^2}$$

Out[45]= $$e\, (a\cdot x)\, (k\cdot X) + \frac{e^2\, \mathrm{Fx}^2\, \mathrm{s1}}{12\, m^2} - \frac{m^2\, x^2}{4\, \mathrm{s1}} - \mathrm{s1}$$

In[46]:=
```
Matrix91 = Expand[Matrix9 /. {DiracGamma[Momentum[FDxv, D], D].GA[5] →
          Contract[DiracGamma[LorentzIndex[μ, D], D].GA[5] × FDxv[μ] /.
             {FDxv[μ_] → -LCD[μ, α1, α2, α3] × xv[α1] × av[α2] × kv[α3]},
           EpsContract → False]} /. antiTripleGamma /.
       {DiracGamma[Momentum[FFx, D], D] → -av2 kx DiracGamma[Momentum[k, D], D]} /.
     {σF → -2 ⅈ DiracGamma[Momentum[a, D], D].DiracGamma[Momentum[k, D], D]}]
```

Out[46]= $$\frac{a^2\, e^2\, \mathrm{s1}\, \gamma\cdot k\, (k\cdot x)}{3\, m^3} + \frac{e\, \mathrm{s1}\, (\gamma\cdot a).(\gamma\cdot k)}{m^2} + \frac{e\, (a\cdot x)\, \gamma\cdot k}{2\, m} - \frac{e\, \gamma\cdot a\, (k\cdot x)}{2\, m} + \frac{e\, (\gamma\cdot a).(\gamma\cdot k).(\gamma\cdot x)}{2\, m} + \frac{m\, \gamma\cdot x}{2\, \mathrm{s1}} + 1$$

In[47]:=
```
Expand[(Matrix91 / (m / 2 / s1) /. {s1 → s * m^2}) / 2 / s]
```

Out[47]= $$\frac{1}{3}\, a^2\, e^2\, s\, \gamma\cdot k\, (k\cdot x) + e\, m\, s\, (\gamma\cdot a).(\gamma\cdot k) + \frac{1}{2}\, e\, (a\cdot x)\, \gamma\cdot k - \frac{1}{2}\, e\, \gamma\cdot a\, (k\cdot x) + \frac{1}{2}\, e\, (\gamma\cdot a).(\gamma\cdot k).(\gamma\cdot x) + m + \frac{\gamma\cdot x}{2\, s}$$

Another representation of the γ - matrix preexponent term

In[48]:=
```
MartixAnother =
  (2 m s1 * 1 / 2 / m / s1 GAD[α] (xv[α] - s1 e Fxv[α] + s1^2 / 3 e^2 FFxv[α]) + 2 m s1 * 1).
    ( 1 + e s1 DiracSlash[a, k, Dimension → D]) /. {s1 → s1 / m^2}
MartixAnother1 = DiracSimplify[Contract[DotSimplify[MartixAnother / (2 s1 / m) /.
       {Fxv[α_] → kv[α] ax - av[α] kx, FFxv[α_] → -av2 kx kv[α]}]]]
CoeffAnother = Coeff9 * m / 2 / s1
```

Out[48]= $$\left(\gamma^{\alpha}\left(\frac{e^2\, \mathrm{s1}^2\, \mathrm{FFx}^{\alpha}}{3\, m^4} - \frac{e\, \mathrm{s1}\, \mathrm{Fx}^{\alpha}}{m^2} + x^{\alpha}\right) + \frac{2\, \mathrm{s1}}{m}\right).\left(\frac{e\, \mathrm{s1}\, (\gamma\cdot a).(\gamma\cdot k)}{m^2} + 1\right)$$

Out[49]= $$\frac{a^2\, e^2\, \mathrm{s1}\, \gamma\cdot k\, (k\cdot x)}{3\, m^3} + \frac{e\, \mathrm{s1}\, (\gamma\cdot a).(\gamma\cdot k)}{m^2} - \frac{e\, (a\cdot x)\, \gamma\cdot k}{2\, m} + \frac{e\, \gamma\cdot a\, (k\cdot x)}{2\, m} + \frac{e\, (\gamma\cdot x).(\gamma\cdot a).(\gamma\cdot k)}{2\, m} + \frac{m\, \gamma\cdot x}{2\, \mathrm{s1}} + 1$$

Out[50]= $$i\, 2^{-D-1}\, e^{-\frac{1}{4} i \pi D}\, \pi^{-D/2}\, \Lambda^{4-D}\, m^{D}\, \mathrm{s1}^{-\frac{D}{2}-1}$$

In[51]:=
```
Matrix91 - Expand[MartixAnother1]
```

Out[51]= $$\frac{e\, (a\cdot x)\, \gamma\cdot k}{m} - \frac{e\, \gamma\cdot a\, (k\cdot x)}{m} + \frac{e\, (\gamma\cdot a).(\gamma\cdot k).(\gamma\cdot x)}{2\, m} - \frac{e\, (\gamma\cdot x).(\gamma\cdot a).(\gamma\cdot k)}{2\, m}$$

Final result for the electron propagator in a CCF

$$S^c(x_2, x_1) =$$

$$\Lambda^{4-D}\int\frac{d^Dp}{(2\pi)^D} E_p(x_2)\,\frac{i((\gamma p)+m)}{p^2-m^2+i0}\,E_p^{bar}(x_1) = e^{-i\frac{\pi}{2}\left(\frac{D}{2}-1\right)}\frac{\Lambda^{4-D}}{2^D\pi^{D/2}}\,m^{D-1}\,e^{i\eta}$$

$$\int_0^\infty\frac{ds}{s^{D/2}}\left[1+\frac{m(\gamma x)}{2s}-\frac{e^2 s(\gamma FFx)}{3m^3}+\frac{ies\left(\sigma^{\alpha\beta}F_{\alpha\beta}\right)}{2m^2}+\frac{ieF_{\alpha\beta}{}^{*}x^\beta\gamma^\alpha\gamma^5}{2m}\right]$$

$$e^{-is-i\frac{m^2x^2}{4s}+i\frac{s}{12}\frac{e^2}{m^2}(Fx)^2} =$$

$$= e^{-i\frac{\pi}{2}\left(\frac{D}{2}-1\right)}\frac{\Lambda^{4-D}}{2^D\pi^{D/2}}\,m^{D-1}\,e^{i\eta}$$

$$\int_0^\infty\frac{ds}{s^{D/2}}\left[1+\frac{m(\gamma x)}{2s}+\frac{e(\gamma a)(kx)}{2m}-\frac{e(\gamma k)(ax)}{2m}+\frac{e(\gamma x)(\gamma a)(\gamma k)}{2m}+\right.$$

$$\left.\frac{es(\gamma a)(\gamma k)}{m^2}+\frac{e^2a^2s(\gamma k)(kx)}{3m^3}\right]e^{-is-i\frac{m^2x^2}{4s}+i\frac{s}{12}\frac{e^2}{m^2}(Fx)^2} =$$

$$= e^{-i\frac{\pi}{2}\left(\frac{D}{2}-1\right)}\frac{\Lambda^{4-D}}{2^{D+1}\pi^{D/2}}\,m^{D}\,e^{i\eta}$$

$$\int_0^\infty\frac{ds}{s^{D/2+1}}\left[\frac{2s}{m}+\gamma^\alpha\left(g_{\alpha\beta}-\frac{es}{m^2}F_{\alpha\beta}+\frac{e^2s^2}{3m^4}F_{\alpha\lambda}F^{\lambda}{}_{\beta}\right)x^\beta\right]$$

$$\left(1+\frac{ies}{2}\sigma^{\alpha\beta}F_{\alpha\beta}\right)e^{-is-i\frac{m^2x^2}{4s}+i\frac{s}{12}\frac{e^2}{m^2}(Fx)^2}$$

$\eta = e(ax)(k,(x_1+x_2)/2)$,

$x = x_2 - x_1$,

$e > 0$,

$\sigma^{\alpha\beta} = \frac{i}{2}\left(\gamma^\alpha\gamma^\beta-\gamma^\beta\gamma^\alpha\right)$,

$\gamma^5 = i\gamma^0\gamma^1\gamma^2\gamma^3$,

$$e^{-i\frac{\pi}{2}\left(\frac{D}{2}-1\right)}\frac{\Lambda^{4-D}}{2^D\pi^{D/2}}\,m^{D-1}\to\frac{(-i)\,m^3}{16\pi^2},\quad D\to 4$$